Chapter 1

Notation

e expression

k lvalue expression

m struct member

i, j integer

D data constructor

s statement

T type

v, x, y variable

Chapter 2

Language Defined Types

A type specifies how to interpret a sequence of bits and the operations that can be applied to those bits.

2.1 Fundamental Types

A fundamental type is a type that cannot be described in terms of other types. The `cweet` language provides the following fundamental types.

- `void` only value is void
- `bool`
- `short`, `int`, `long`, `llong` and sized types `i8`, `i16`, `i32`, `i64`, `i128` signed arithmetic
- `ushort`, `uint`, `ulong`, `ullong` for modulo arithmetic and their sized counterparts `u8`, `u16`, `u32`, `u64`, `u128`
- `byte`, `word`, `b8`, `b16`, `b32`, `b64`, `b128` bitwise operations
- `char`, `wchar`, `c8`, `c16`, `c32` characters
- `float`, `double`, `ldouble`, `ddouble`, `f8`, `f16`, `f32`, `f64`, `f128` floating point

The result of overflow or underflow on signed integers is implementation defined. Use `uint` for modulo arithmetic. Use sized types only when there is a special need for it. Use `double` when a floating-point number is needed.

2.2 Complex Types

A complex type is a type that is defined in terms of other types. More advanced programming languages provide facilities to define complex types called generic types. In `cweet` the following are the only complex types.

- `[n]T` An array of `n` elements of type `T`.

- `[]T` A slice of elements of type `T`.
- `&T` A non-nullable pointer to an object of type `T`.
- `*T` A nullable pointer to an object of type `T`.

2.2.1 Arrays

An array `[n]T` consists of `n` elements of type `T` laid out contiguously in memory. If `T` is an array type, then the type `T` is called a multi-dimensional array. A type `[a1][a2]...[ak]T` where `T` is not an array type is a `k`-dimensional array with dimensions `a1,...,ak` and base type `T`. It can also be viewed as a 1-dimensional array of dimension `a1` where each element is of type `[a2]...[ak]T`.

Arrays support assignment, indexing, and slicing.

If `a : [a(1)]...[a(k)]T`, then an index expression `a[i(1)]...[i(m)]` refers to the element of type `[a(m+1)]...[a(k)]T` allocated as part of `a` where `m <= k` and `0 <= i(j) < a(j)` for all `j`.

The expression `a[]` produces a `k`-dimensional slice with the same dimensions as `a` that refers to the elements of `a`. The expression `a[i..j]` produces a `k`-dimensional slice that refers to rows `i` through `j` of the array `a`.

2.2.2 Slices

A `k`-dimensional slice with base type `T` is the following type.

```
struct {
    ptr : &T,
    len : [k]int
}
```

and refers to a contiguous block of memory holding elements of type `T`.

A `k`-dimensional slice can be sliced further to create another `k`-dimensional slice. The expression `s[i(1)]...[i(m)]` produces an `k-m` dimensional slice obtained by fixing the first `m` indices to these values. If `m = k`, this expression refers to the underlying value. The expression `s[i..j]` produces a `k`-dimensional slice that refers to the contiguous block of memory referred to by `s[i]` through `s[j]`.

The following function in `std.slice` can be used to reshape slices.

```
fn reshape (
        in : slice(#n, ?t),
        dims : [#m]int
   ) : slice(#m, ?t);
```

This produces an `m` dimensional slice from an `n` dimensional slice. The total number of elements have to be the same.

Slices of slices can be used to create jagged arrays. Unlike multidimensional slices, the base elements need not be contiguously laid out in memory. For example, the type `[]([]int)` is a slice of slices of `int`s.

2.2.3 Pointers

The type `&T` (`pointer(T)`) can never be `Null`. This allows the compiler to elide null pointer checks.

The types `*T` (`npointer(T)`) can take `Null` as a value and therefore the expression `*a` can fail.

2.3 User Defined Types

The **type** keyword is used to define types.

An **enum** is a restriction of values of one of the integral types.

```
type Colour = enum {
  red,
  green,
  blue
};
// Use integral values for constants. Like C enum.
type Colour = enum(int) {
  red = 1,
  green,
  blue
};
// Use a different integral type.
type Colour = enum(long) {
  red, green, blue,
  white, black, gray,
};
```

enum types are not interchangeable with integral types unless specified in the declaration. By the same rule, you cannot specify values of enum members as integers unless you specify a base integral type. An enum with a base integral type has all the operations of the base integral type.

The **struct** keyword is used to define product types. All members of a struct are labelled. The layout is the same as a **struct** in `C`.

```
type Point2D = struct { x : int, y : int };
```

A **struct** with unlabelled members is called a tuple. The members are implicitly labelled `a, b, ....`

```
type Point3D = {int, int, int};
// equivalent to
type Point3D = struct {
    a : int,
```

```
    b : int,
    c : int
};
```

The **union** keyword is the same as `C` **union**.

The **choice** keyword can be used to define tagged or discriminative unions. A **choice** has a finite number of choices and data members in each of those choices.

```
type NetStatus = choice {
  Connected{ip : [4]byte},
  Disconnected
};
```

The data members in a **choice** (For example, `ip`) can be left unnamed.

An anonymous discriminative union is the sum analogue of tuples.

```
type IoF = int | float;
```

Any value of type present in an unlabelled choice can be implicitly converted to a value of the unlabelled choice type. For example, it is valid to initialize or assign an object of type `IoF` from an `int` or a `float`.

Anonymous sums have the following properties.

- (absorption) `T|T = T`.
- (commutativity) `T|U = U|T`
- (associativity) `T|(U|V) = (T|U)|V = T|U|V`
- (identity) `T|`**unreachable** `= T`

2.3.1 Abstract Types and Type Aliases

A type definition **type** `T = U;` creates an abstract type `T` that has the same representation as `U`. If `x: T`, then the expression `x` **is** `U` has type `U`. On the other hand, if `y:U`, we have `T{y}: T`.

The declaration `alias T = :U;` specifies that the name `T` is an alias for the type `U`. It does not create new types.

2.4 Aggregate Types

Arrays, slices, structs, tuples, unions, and tagged and anonymous choice types are all aggregate types. Each aggregate type is a collection of member types. For example, for slices, the members are a pointer to the underlying type and an `int`.

2.5 Type Conversions

The standard module `std.conv` provides functions that implement explicit type conversions supported by the language.

TODO: Define allowed conversions.

2.6 Properties of Types

Table 2.1 lists the core properties satisfied by various types. A new type constructed by a **type** declaration can derive these properties using an **is** clause. For example,

```
type Length is (Numeric, Eq, Ord) = int;
```

Property	Operations	Types
Numeric	`(+, -, *, /): T -> T -> T` `-: T -> T`	`int, float`
Integral	`%: T -> T -> T`	`uint, byte`

Bitwise	`(&, |, ^): T -> T -> T`
	`~: T -> T`
	`(<<, >>, >>>): T -> int -> T`
		`byte`

Logical	`(&&, ||): T -> T -> T` `!: T -> T`	`bool`
Index	Usable as index	`int`
Ord	`(<, >, <=, >=): T -> T -> bool`	`float`
Eq	`(==, =/=): T -> T -> bool`	`pointer`
Type		All Types
Abstract	Data Hiding	
Same	All operations of underlying type	

Table 2.1: Core Properties of Types

2.7 Implicit Conversions

We write `X` $\rightarrow$ `Y` for types `X` and `Y` to specify that `X` is implicitly convertible to `Y`.

1. Enums with integral types `I` are implicitly convertible to `I`.
2. The value **error** is implicitly convertible to `Null`.
3. All other values are not convertible to **error**.
4. `X` $\rightarrow$ `(X'|Y)` if `X` $\rightarrow$ `X'` and not `X` $\rightarrow$ `Y`.
5. `X|Y` $\rightarrow$ `X'` if `X` $\rightarrow$ `X'` and `Y` $\rightarrow$ `X'`.

Chapter 3

Statements and Expressions

A statement produces no useful value and is specified for potential side-effects. An expression produces a value that can be used by other expressions or statements. In `cweet` many constructs are available as both statements and expressions. For example:

```
// if-else expression.
var max = if (a > b) a else b;

// if-else statement.
var max : int = undefined;

if (a > b) max = a;
else max = b;
```

On the other hand, constructs such as **return** statements and assignment statements are not available as expressions.

3.1 Variable Declarations

A variable declaration is a statement such as:

```
var i : int = 42;
```

does the following things:

1. Allocates storage for an `int`.
2. Binds the identifier `i` to the storage.
3. Initializes the storage location with the representation for value `42`.

For user-defined types, one can define variables as follows:

```
// For product types use the typename
var p = Point2D{0, 0};
var q = Point2D{.x = 1, .y = 1};
var r = Point3D{0, 0, 0};

// For tuples, use a tuple literal.
var x = {0, 1};

// For choice types use the constructors.
var s = NetStatus.Connected{[127, 0, 0, 1]};

// For unlabelled choices, infer constructor.
var x : IoF = 42; // int

// For enums, use one of the members.
var bg = black;
var fg = Colour.white;
var border : Colour = red;
```

When defining variables, an `_` can be used to ask the compiler to infer parts of the type specification.

The special value **undefined** can be used to leave a storage location uninitialized. An `_` can be used in place of the variable name if the variable is not going to be used.

3.2 Literals

3.2.1 Numeric Literals

Integral literals are polymorphic and default to `int`. Floating point literals are also polymorphic and default to `double`. Underscore can be used as a digit separator. The prefixes `0x` and `0b` can be used for hexadecimal and binary format. The character `p` can be used for scientific notation.

```
100_000
0xcad
0b101010
6.023p23
1p-6
0b1.110p+3
```

A numeric literal cannot be followed by an alphanumeric, underscore, or a period.

3.2.2 String Literals

String literals are stored as a null-terminated array of `char` in the data section. Their type is `[]char` in `cweet` and the null-terminator is ignored by the slice.

A string enclosed within `"` is a string literal with escape sequences. A raw string literal is written as `r"c<Contents␣here>c"`. The character `c` can be any character except `"` and acts as part of the delimiter.

```
  "Hello,␣world\n"
  "utf8␣\u8{0d09}␣utf16␣\u16{0d09}␣utf32␣\u32{0d09}"
  "Embedded␣null␣bytes␣\x{0000}"
  r"␣Hello,␣world␣" // space as delimiter.
  r"|
␣␣␣␣␣int␣main()
␣␣␣␣␣{
␣␣␣␣␣␣␣␣␣␣return␣0;
␣␣␣␣␣}
␣␣|" // Everything between two |'s.
```

3.2.3 User Defined Type Literals

```
Point{.x = 9, .y = 1} // struct
{.x = 9, .y = 1} // struct, type inferred
Network.Connected{[127, 0, 0, 1]} // choice
Connected{[127, 0, 0, 1]} // choice, type inferred.
Network.Disconnected // choice
Colour.red // enum
red // enum, type inferred
```

3.2.4 Symbols

A symbol literal is an identifier preceeded by a backquote. A symbol stands for itself and symbols are available only at compile-time. These are used as **loop** labels and for introducing variable bindings.

3.3 Assignment

An assignment statement mutates some storage location. The simplest form `lexpr = expr;` mutates the contents of the storage location referred to by `lexpr` using the value of `expr`. The assignment operation is defined for all types as a simple `memcpy`.

For infix operators `op` except for comparison operators, the augmented assignment `a op= b;` is same as in `C`.

The left-hand side of an assignment has to be an lvalue expression. The following characterizes lvalue expressions:

1. A variable declared using **var** is an lvalue.
2. k.m where k is a struct and m is a member.
3. k.D where k is a choice for which D is a constructor.
4. *e for both nullable and non-nullable pointer types.
5. k is T
6. k[i]
7. A variable binding in a match expression if the expression being matched against is an lvalue.
8. k!

3.4 Comparison Operations

The operators `==, =/=, <, >, >=, <=` are available. They produce Boolean values. The equality operators are defined for all fundamental types. They are not defined for user defined types because of padding. The comparison operators are defined for all numerical types includine `byte` etc.

3.5 Arithmetic Operations

Usual binary arithmetic operators `+, -, *, /, %` are supported. The unary `-` evaluates to the additive inverse in the ring. The unary `+` is identity and is provided for consistency.

3.6 Boolean Logic

The operators `&&, ||, !` are used to combine Boolean expressions. In addition to comparison operations, the following expressions also produce Boolean values.

```
type Colour = enum { red, green, blue };
var bg = red;
assert (bg.red && !bg.green);
```

3.7 Conditionals

A conditional can be an expression or a statement.

If `expr` has type `T`, then **when**`(condn) expr` has type `T|void`. In the statement **when** `(expr) stmt`, `stmt` is executed only if `expr` evaluates to `true`.

If `expr1` and `expr2` has types `T` and `U`, then **if**`(condn) expr1` **else** `expr2` has type `(T|U)`.

An **ensure** produces `T|`**error**.

```
(ensure (x != 0) 1/x) : double|error
```

Similarly, a case expression has type that is the anonymous discriminative union of its arms' types. If the catch-all pattern `_` is absent, then the type union contains `void`.

```
fn abs(n : int) : int
{
    return case {
        n >= 0 => n,
        _      => -n,
    };
}
```

Therefore, the following fails to compile.

```
fn abs(n : int) : int
{
    return case {
        n >= 0 => n,
        n < 0  => -n,
    };
}
```

The values **unreachable** and **error** have types **unreachable** and **error**.

The following demonstrates how to use these with **case** expressions.

```
return case {
    n > 0 => n,
    _     => -n,
} : int

return case {
    n > 0 => n,
    n < 0 => -n,
} : int|void

return case {
    n > 0 => n,
    n < 0 => -n,
    _     => unreachable,
```

```
} : int|unreachable -> int

return case {
    n > 0 => n,
    n < 0 => -n,
    _     => error,
} : int|error
```

3.8 Loops

The **loop** expression and statement is inspired by Lisp's loop macro. The general form is:

```
loop
'label
for (var x; slice)
while (bexpr1)
until (bexpr2)
break (val)
stmt
```

The **with** clause is used to setup variables only used within loop's scope. The **for** clause specifies iteration variables and what to iterate over. The **until** and **while** clauses specify stopping criterion. If the loop is an expresion, then the **break** clause specifies the value of the expression. The body of the loop may also contain **break** statements, optionally labelled, that may or may not return a value. The body of the loop must be enclosed in braces.

```
var i = loop for(var i; 1000 ..)
             until (is_prime(i)) break(i) {};
```

assigns the smallest prime larger than `1000` to `i`.

A **continue** statement can be used to skip to the next iteration of the loop. A label can be specified to identify the loop.

3.9 Lexical scoping

The **do** keyword can be used to specify that a block of code delimited by braces is an expression. The type of this expression is the type of the last expression in the block. (This keyword is used to keep the grammar LL(1) by avoiding conflicts with tuple/struct literals).

```
var i = do {
    // Complex initialization.
```

```
    e // expression used to initialize i
};
```

A statement beginning with an open brace is treated as the start of a lexical scope.

```
{
    var a = 0;
    print(a);
}

{0, 0}; // Error.
({0, 0}); // Ok
```

On the other hand, when expecting an expression, the compiler treats an open brace as the beginning of a struct literal. The **do** keyword can be used to specify that the following open brace indicates the start of a block expression and not a struct literal.

The **with** keyword introduces a new lexical block along with some statements at the beginning of the block. This generalizes initializing expressions in for-loops in `C` and conditional statements in `C++`. The value of the following expression is the last element of the list `head`.

```
with (
    var p = head;
    var q = head;
) loop while (p != Null) break(q) {
    q = p;
    p = p.next;
}
```

A **with**`(...)` can be followed by a statement, expression (If **with** is an expression), or a sequence of statements enclosed between braces.

3.10 Dereferencing Operations

Pointers can be dereferenced using the unary prefix operator `*`. A `*T` dereferences into a `T|`**error** and a `&T` dereferences into a `T`.

3.11 Pattern Matching

A **match** statement can be used to pattern match against many types.

```
// int
match (i) {
```

```
    `j == .. 100 => j,
    _            => @max(:int),
}

// arrays
match (a) {
    [`i == 0 .. 100, ...]      => i,
    [_, `j == 0 .. 100, ...] => j,
    _                          => 1000
}

// enum
match (day) {
    // range of values.
    Tuesday .. Thursday => "Midweek",
    // set of values.
    {Monday, Friday}    => "Bridgeday",
    _                   => "Weekend",
}

// choice
match (status) {
    Connected{[127, 0, 0, 1]}   => "localhost",
    Connected{[255, 255, ...]}  => "localnet",
    Connected{`a .ip}           => format_ip(a),
    Disconnected                => "disconnected"
}

// tuple
match (point) {
    {0, 0}           => origin,
    {0 .. 5, 0 .. 5} => near_origin,
    _                => somewhere
}

// struct
match (point) {
    // Use field names.
    {.y == 0, .x == 0}     => origin,
    {0 .. 5, 0 .. 5}       => near_origin,
    _                      => somewhere
}

// Anon choice
match (iof) {
    `x: int   => x + 1,
```

```
    `x: float => x + 0.1,
}

// pointers
match (p: *int) {
    Null => "Nope.",
    _    => with (var q = p->ptr()!) stringify(q),
}
```

The general form of a pattern to match against a field is

```
`id .field :type == pat
```

This binds the named field to the identifier `id` and the pattern succeeds only if value of `field` matches `pat`. The **type** is used when matching against anonymous choices.

The type of a **match** expression is assigned using the same rules as for **case**.

A match expression can also match against multiple expressions.

```
match (p1, p2) {
    (
        {0 .. 10, 0 .. 10},
        {0 .. 10, `y}
    ) => y += 10;
}
```

Note that the above will not work by matching against `{p1, p2}` because `y` will not be bound to an lvalue. A **pmatch** expression can be used when only one match arm is used.

```
pmatch (
    p1 as {0 .. 10, 0 .. 10},
    p2 as {0 .. 10, `y}
) y += 10;
```

An **is** expression can be used to select a particular choice from an anonymous discriminative union.

```
var x : int|bool = 42;
var y = (x is int) + 1;
```

Here the expression `x is int` has type `int|error`.

3.12 Array and Slice Operations

Arrays and slices have member `len : int` that gives the number of elements in the array. A slice can be obtained from an array or a slice using a slice expression.

```
// primes has type [6]int
var primes = [2, 3, 5, 7, 11, 13];
var a = primes[..];
var b = primes[1 .. 4];
var c = primes[.. 3];
var d = primes[1 ..];
```

A range expression `beg .. end` specifies a closed range of numbers or enum values. In a slice expression, the first element defaults to `0` and the last one defaults to `len - 1`. The inequalities `beg >= 0` and `end >= beg - 1` must hold when slicing. These rules allow using `a[0 .. -1]` to construct a zero length slice to the beginning of `a`.

The type `[][]T` is a multidimensional slice. The type `[]([]T)` is a slice of slices. A multidimensional slice points to a contiguous block of memory.

3.13 Bitwise Operations

The infix operators `>>`, `<<`, `^`, `~`, `&`, `|` are the usual bitwise operators. The infix operator `>>>` is the arithmetic shift right.

3.14 Piping Operator

The `->` operator is syntactic sugar for function calls that states the flow of data clearly. For example,

```
f(g(h(x, y), z), u);
// is the same as
x->h(y)->g(z)->f(u);
```

3.15 scope statements

`scope(exit)` statement allows statements to be executed on function exit.

`scope(success)` statement allows statements to be executed on successful function exit.

`scope(failure)` statement allows statements to be executed on error function exit.

3.16 Precedence and Associativity

Below are the operators from the ones that bind tightest.

Pointers to structures and members that are pointers to structures are common. For example, given a pointer `x` to a node with `next` and `prev` pointers,

Precedence	Operator	Associativity	Examples

Term	Variables
	Literals
	Parenthesis
		NA	`x`
			`2`
			`"Hello, world!"`
			`[2, 3, 5, 7]`
			`{.x = 1, .y = 2}`
			`(x+y)`

Access	`.`	Left	`x.m`

Postcircumfix	Function Call
	Indexing/Slicing
	Construction
		Left
		Left
		NA
			`x.f()`
			`x.a[i]`
			`T.D{xs}`

Pipe	`->`	Left	`x->f()`

Unary	Dereference
	Address
	Negation
	Logical NOT
	Complement
		Right
		Right
		Right
		Right
		Right
			`*f() (*x)->f()`
			`&a[i]`
			`-f()`
			`!f()`
			`~f()`

Choose	**is**	Left	`(x->f() is int) + 1`
Bang	`!`	NA	`x->f() is int!`
Annotate	`:`	NA	`4 * 2 : float`
Mul	`* / %`	Left	`n * f()!`
Add	`+ -`	Left	`a + b*c`
Shift	`<< >> >>>`	Left	`a+b >> 3`
BitAnd	`&`	Left	`a<<1 & b`
BitXor	`^`	Left	`a ^ b&c`
BitOr	`|`	Left	`a^b | c`
Ord	`< > <= >=`	NA	`a|b < b|c`
Eq	`== =/=`	NA	`a<b == b<c`
And	`&&`	Left	`a<b && b<c`
Or	`||`	Left	`a && b || c`
Stmt	**if else** **when** **ensure** **case** **loop** **match** **pmatch**		**if** `(b) c` **else** `d || e`
Lambda		NA	`\b =>` **if** `(b) 1` **else** `0`

expressions like `(*(*x).next).prev` may be necessary. This is unwieldy. As an exception, pointers to aggregates such as struct, choice, enum, tuples, slices (for `ptr` and `len`), and arrays (for `len`) are automatically dereferenced when performing member access using the `.` operator allowing one to write `x.next.prev`. The other exception is that pointers to functions are automatically dereferenced when calling the function. i.e., the code `(*f)()` and `f()` are equivalent.

Chapter 4

Functions

Functions are defined using **fn** keyword. The type of a function includes the types of parameters and its return type. All functions are known at compile-time and a variable cannot have **fn** type. But pointers to functions are allowed.

Nested functions are allowed and mainly acts as a namespacing mechanism. If the nested function accessess the outer function's local variables, then a pointer to the nested function cannot be returned from the function, cannot be stored into variables, and a pointer to it cannot be passed to other functions that do.

4.1 Tail-call Elimination

A **become** statement is the same as **return** except that it is guaranteed not to push an additional stack frame. This is only valid if the called function is explicitly named i.e., the tail-call is not through a function pointer.

4.2 Lambda Functions

Lambda functions are allowed in some contexts where there is no need for dynamic allocation to use the lambda. This is possible if the lambda doesn't access any variables from its environment or all calls to the lambda function can be inlined in the context its capturing. For example, when it is passed as a function pointer to a function that can be inlined and the called function only uses the function pointer to call it and not store it into other variables.

These rules also apply to nested functions.

```
fn map(x : []int,
       f : &fn(int) : int) : void {
    loop for (var e; x) {
        e = (*f)(e);
```

```
    }
}

fn foo() : void {
    var a = [1, 2, 3];
    var sum = 0;
    map(a[], \i => do { sum += i; i });
}

// compiled into
fn foo() : void {
    var a = [1, 2, 3];
    var sum = 0;
    loop for (var e; a[]) {
        e = do { sum += e; e };
    }
}
```

Chapter 5

Error Handling

Errors are represented by the special value **error** of type **error**. If the execution of a program encounters the value **error**, then the program does an error-return from the current function or the value **error** should be stored to a variable capable of holding this value.

Lets start with an example. Consider the following function that may err:

```
fn safediv(x, y: int): int|error
{
    return ensure(y =/= 0) x/y;
}
```

A function calling this function has the following choices.

```
var x: _|error = safediv(4, 0); // Store error.
var x = match safediv(4, 0) { // Handle error.
     :error  => 0,
    `x: int  => x,
};
var x = safediv(4, 0); // Propagate error.
```

An error can propagated if the error value occurs in a function that returns one of the following types.

1. `T|error` for some type `T`.
2. `*T` for some type `T`. Here the value `Null` corresponds to error.

An error can be ignored using the postfix bang operator. The expression `e!` has type `T` whenever the expression `e` has type `T|error`.

Chapter 6

Packages

Packages provide a simple namespacing mechanism for the symbols in your program. A package corresponds to a directory or file.

Given a package named `foo.bar`. The compiler looks for the following folder structure.

```
foo
 |
 |-- mod.cw
 |
 |-- bar.cw
```

where `foo/mod.cw` contains the following code.

```
package foo (bar, ...);
package bar;
```

and `foo/bar.cw` contains

```
package bar (...);
```

The package declaration lists all symbols exported by the package. These symbols can be types, functions, interfaces, modules, or packages.

The contents of a package can be imported into the current scope by the following declaration:

```
from path.to.package as P
    import (s1, s2, ...);
```

Chapter 7

any and Polymorphism

The pointer types `&any` `*any` play the role of `void *` in `C`. The cast operator `any_cast!(*T)(ptr)` from the module `std.any` can be used to convert from or to such pointers. It is undefined behaviour to convert a `*any` to a pointer that was not the original type.

Chapter 8

`repr` and Type Aliasing

The `repr` type plays the role of `std.byte` in `C++`. Any object can be converted into a `[]repr` by using the function `repr` from `std.repr` module. This can be used to reinterpret the bits of one object as another.

A pointer to `repr` is the only pointer type that supports pointer arithmetic. The following operations are supported.

1. `(p : &repr + i : int) : &repr` points to `i` bytes after `p`.
2. `(p : &repr - q : &repr) : int`. The pointers `p` and `q` has to point to the same memory block. A pointer to one element past the allocated array is considered a pointer to the memory block.

TODO: Allow sane type aliasing and annotations to specify when pointers can and cannot alias.

Chapter 9

Parsing Oddities

```
{f(), g()}; // Syntax error. Statement block.
({f(), g()}); // Ok. Tuple expression.

x: int | y // x annotated with type int|y
(x: int) | y // Correct

// In patterns
`x: int .x == 5
(`x: int) .x == 5
```